
Solution to small exercises: 

libname asb 'C:\Documents and Settings\xx12345\Desktop'; 

 

* Assignment 1; 

proc means data=asb.fltattnd clm mean std; 

var salary; 

run; 

 

* Assignment 2  *; 

proc print data=asb.weekrev; 

var FlightID origin Date CargoRev PasRev; 

run; 

 

* Assignment 3a) *; 

proc sort data=asb.sales121999; 

by FlightID; 

run; 

 

*Assignment 3) *; 

data total_rev;  

set asb.sales121999; 

Total = sum(FClassRev, BClassRev, EClassRev, CargoRev); 

keep FlightID RouteID Total; 

run; 

proc print data= total_rev; 

run; 

 

*Assignment 4) *; 

data new_pilots; 

set asb.pilots; 

if JobCode='PT2' and Salary gt 84000 then output; 

run; 

proc print data=new_pilots; 

var IDNum Salary ; 

run; 

 

*Assignment 5) *; 

Proc reg data=asb.b_fitness; 

Model oxygen_consumption = runtime age performance weight; 

Run; 

quit; 

Event study example: 

/*Read in the data using a data step using the import function*/ 

 

data event; /*name of new dataset*/ 

set event; /*name of old which will be overwritten*/ 

count = _N_; /*adding a variable 'count' that is the observation number*/ 

run; 

 

/*new dataset namede event200 with only the first 200 observations from the 

dataset event*/ 

data event200;  

set event; 

where count <= 200;  

run; 

 

/*market model for the 5 stocks in the model the estimates are written to the 

sas dataset beta*/ 

proc reg data=event200 outest=beta noprint; 

model stock1 = market1; 

model stock2 = market2; 

model stock3 = market3; 

model stock4 = market4; 

model stock5 = market5; 

run; 

quit; 

 

/*the following is made to make the alpha and beta estimates ready to merge with 

the event dataset so the abnormal returns can be calculated*/ 

data beta1; /*the estimates will be saved in the dataset beta1*/ 

set beta; 

if _n_=1 then do; 

beta1= market1; 

alpha1= intercept; 

end; 

if _n_=2 then do; 

beta2= market2; 

alpha2= intercept; 

end; 

if _n_=3 then do; 

beta3= market3; 

alpha3= intercept; 

end; 

if _n_=4 then do; 

beta4= market4; 

alpha4= intercept; 

end; 

if _n_=5 then do; 

beta5= market5; 

alpha5= intercept; 

end; 

keep beta1-beta5 alpha1-alpha5; 

run; 

 

/*to make all estimates apear in one row the follwoing means procedure is used 

(this is a small trick)*/ 

/*the estimates will be in the dataset beta2*/ 

proc means data=beta1 mean noprint; 

output out=beta2; 

run; 

data beta2; 

set beta2; 

where _stat_='MEAN'; 

drop _type_ _freq_ _stat_; 

run; 

 

data abnorm; /*new dataset called abnorm*/ 

set event; /*the dataset is based on the event dataset i.e. data from both the 

estimation and event period*/ 

if _n_=1 then set beta2; /*all the alpha and beta estimates are added to the 

dataset*/ 

run; 

 

/*the abnormal returns are calculated in the dataset abnorm by using the market 

model estimates*/ 

data abnorm; 

set abnorm; 

abnor1 = stock1 - alpha1 - beta1*market1; 

abnor2 = stock2 - alpha2 - beta2*market2; 

abnor3 = stock3 - alpha3 - beta3*market3; 

abnor4 = stock4 - alpha4 - beta4*market4; 

abnor5 = stock5 - alpha5 - beta5*market5; 

run; 

 

/*first the average excess return is calculated for each day in the event and 

estimation window (slide 81)*/ 

data a_bar; 

set abnorm; 

a_bar = mean(of abnor1 - abnor5); 

run; 

 

/*calculate a_bar_bar (which is the sum of a_bar in the estimation period 

divided by the number of observations i.e. average a_bar)*/ 

/*The a_bar_bar estimate is now in the dataset a_bar_bar*/ 

proc means data=a_bar noprint; 

where count<201; /*only in the estimation period*/ 

var a_bar; 

output out=a_bar_bar mean=a_bar_bar; 

run; 

 

/*calculating std of a_bar*/ 

/*first the squared deviations from a_bar_bar is calculated*/ 

data std_a_bar; 

if _n_=1 then set a_bar_bar; /* a_bar_bar is added to the table*/ 

set a_bar; 

sqrd = (a_bar - a_bar_bar)**2; 

run; 

 

/*next we sum the sqared deviation from the mean*/ 

proc means data=std_a_bar noprint; 

where count<201; /*only in the estimation period*/ 

var sqrd; 

output out=sum_sqrd mean=sum_sqrd; 

run; 

 

/*lastly the variance used in the calculatation of the test statistics is 

calculated*/ 

data sum_sqrd; 

set sum_sqrd; 

std_a_bar = sqrt(sum_sqrd); /*std used in the T1 test statistic*/ 

std_car = sqrt(3*(std_a_bar**2)); /*this esimate is used later see slide 84*/ 

run; 

 

/*calculating the T1 test statistic in each day of event window*/ 

data event_window; /*new dataset with only observations from the event window*/ 

set a_bar; 

where count gt 200; 

if _n_=1 then set sum_sqrd; /*the variance is added to the dataset*/ 

t1 = a_bar/std_a_bar; 

pvalue_t1 = 2*(1- cdf('t',abs(t1), 199)); 

run; 

 

/*print the t1 test statistics*/ 

proc print data=event_window; 

var t1 pvalue_t1; 

run; 

 

/*calculating the T1_car test statistic*/ 

/*first the sum of a_bar is calculated*/ 

proc means data=event_window noprint; 

var a_bar; 

output out=car SUM=car; 

run; 

 

/*next the variance calculated above is added to the dataset*/ 

data car; 

set car; 

set sum_sqrd; 

t1car = car/std_car; /*the value of the test statistic can then be found in the 

dataset sum_a_bar2*/ 

pvalue_t1car = 2*(1- cdf('t',abs(t1car), 199)); 

run; 

 

/*print the t1_bar test statistics*/ 

proc print data=car; 

var t1car pvalue_t1car; 

run; 

 


